
1 Corrections required

1.1 f_TC_22_4_22
	Function name
	f_TC_22_4_22

	Reason for change
	1. At Step 1 the power level changes will result in UE reselecting to Cell5, therefore the SRB and DRB configured are not required any more and needs to be released. This is needed to avoid the same EPS bearer context mapped to different cells at the same time. However, SRB1, SRB1bis and DRB should be configured on Ncell1 to be ready for Step 17.

2. It is observed that for change 3 of TTCN CR R5s180217, the MCC160 implementation of resetting the SRBs and DRBs only at the end does not work. For UEs that are retransmiting the RRC Connection Resestablishment Request, the local end configuration for SRBs and DRBs must be reset each time.

3. The TAU procedure on Ncell 5 implicitly assumes that the DRB is already created, whereas, the DRB gets created after the TAU procedure on Ncell 5 and hence when local end release of SRBs and DRBs is called, local end configuration issues occur as far as the release of the DRB is concerned as there is no DRB configured at this point in time

4. The TAU procedure on Ncell 1at Step 23 releases the configured DRB on Ncell 1. Hence, care must be taken to ensure the DRB is not released again during the Postamble as this would result in local end configuration issues.

	Summary of change
	1. It is recommended to do a release of stored security context and perform release of SRB1/1bis and DRB. Also, configure SRB1 /1bis and DRB to be used at Step 17.

2. The function f_NBIOT_SS_SRBs_DRBs_Reset is called for every transmission of the RRC Connection Reestablishment Request.

3. A flag is introduced in the TAU procedure to ensure DRB is not released when performing local end configuration. This also is applicable for change 4.

	TTCN module
	

	MCC160 Comment
	


Before Change:

	function f_TC_22_4_22_NBIOT() runs on NBIOT_PTC

  {

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT3;

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var EUTRA_ASN1_PhysCellId_Type v_PhysCellId;

    var template (present) EUTRA_ASN1_ReestabUE_Identity_Type v_UE_Identity;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var boolean v_IPv4AllocationViaNasFlag := false;

    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var template (value) PDN_Address v_PDN_Address;

    var NBIOT_PDN_Type v_PDN;

    var GutiParameters_Type v_GutiParams;

    var template (value) MobileIdentity v_Guti;

    timer t_Wait;

    v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell5, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_NB_CellPower(nbiot_Cell5, -85)

    };

    v_CellPowerList_AtT3 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell5, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    f_NBIOT_Init(c2, USER_PLANE);  // @sic R5-176139 sic@

    //Set the value of T311 timer in SIB2 for Ncell 1, Ncell 5

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500; // @sic R5-176139 sic@

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms30000;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell5);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500; // @sic R5-176139 sic@

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms30000;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell5, v_SIB2);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell5, USER_PLANE); // @sic R5-176139 sic@

    //Bring UE to initial state @sic R5s170536 sic@

    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);

    // Switch on UE

    f_UT_SwitchOnUE(UT, false);

    v_GutiParams := f_NBIOT_CellInfo_GetGuti(nbiot_Cell1);

    v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    v_NAS_Ind := f_NBIOT_InitialRegistration_Step2_4 (nbiot_Cell1, USER_PLANE, PREAMBLE);

    f_NBIOT_InitialRegistration_Step5_12(nbiot_Cell1, USER_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NAS_Ind);

    v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn(0);  // This is registration, so this is the first PDN

    v_PcoFromUE := v_PDN.protocolConfigurationOptions;

    if (isvalue(v_PcoFromUE)){

      v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);

    }

    v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE);

    v_PDN_Address := f_CIOT_GetPDNAddress(v_PDN.assignedPdnType, v_IPv4AllocationViaNasFlag, -, pdnTypeAsSupportedByUE); // @sic R5s170597 sic@

    v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cs_ATTACH_ACCEPT_Common(tsc_EAttachType_EpsOnly,

                                                                                cs_GprsTimer_v_deact,

                                                                                cs_TAIListNonConsecutive_lv(f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity), {bit2oct(f_NBIOT_CellInfo_GetTAC(nbiot_Cell1))}),

                                                                                v_Guti,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                cs_EPS_NwkFtSup_UP,

                                                                                f_GetNBIOTAdditionalUpdateResult(v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType, USER_PLANE),

                                                                                omit,

                                                                                omit),

                                                        cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                   v_PDN.pti,

                                                                                                   v_PDN.accessPointName,

                                                                                                   v_PDN_Address,

                                                                                                   v_PcoToUE,

                                                                                                   omit,

                                                                                                   omit));

    // Send specific message contents in RRC Connection Reconfiguration message

    SRB.send(cas_NB_SRB1_RrcNasPduList_REQ(nbiot_Cell1,

                                           cs_TimingInfo_Now,

                                           cs_RRCConnectionReconfiguration_NB(tsc_RRC_TI_Def, cs_RadioResourceConfigDedicated_NB_DrbConfig({ cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) },

                                                                                                                                           -,

                                                                                                                                           cs_RLF_TimersAndConstants_Setup_UP)),

                                           {v_NAS_MSG_Request}));

    SRB.receive(car_NB_SRB_RrcPdu_IND(nbiot_Cell1, tsc_SRB1, cr_RRCConnectionReconfigurationComplete_NB(tsc_RRC_TI_Def)));

    // Now finish preamble

    f_NBIOT_NAS_AttachComplete_UP(nbiot_Cell1, tsc_EpsDefaultBearerId);

    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS changes Ncell 1 levels according to row "T1" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //Wait for 10 s.

    f_Delay(10.0);

    //@siclog "Step 3" siclog@

    //The SS changes Ncell 2 levels according to row "T2" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 4-9" siclog@

    f_NBIOT_TrackingAreaUpdate(nbiot_Cell5, USER_PLANE);

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "Step 10" siclog@

    //Wait for 5 s.

    f_Delay(5.0);

    //@siclog "Step 11" siclog@

    //The generic test procedure in TS 36.508 subclause 8.1.5A.2.3 is performed on Ncell 2

    f_NBIOT_GenericRbEst(nbiot_Cell5, mt_Access, -, -, cs_RLF_TimersAndConstants_Setup_UP); // @sic R5s170885 sic@

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionReconfiguration-NB message on Ncell 2

    SRB.send(cas_NB_SRB_RrcPdu_REQ(nbiot_Cell5,

                                   tsc_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_RRCConnectionReconfiguration_NB(tsc_RRC_TI_Def,

                                                                      cs_RadioResourceConfigDedicated_NB_DrbConfig(omit,

                                                                                                                   omit,

                                                                                                                   cs_RLF_TimersAndConstants_Release))));

    //@siclog "Step 13" siclog@

    SRB.receive(car_NB_SRB_RrcPdu_IND(nbiot_Cell5, tsc_SRB1, cr_RRCConnectionReconfigurationComplete_NB(tsc_RRC_TI_Def)));

    //@siclog "Step 14" siclog@

    //The SS changes Ncell 2 levels according to row "T1" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 15" siclog@

    //Wait for 10 s.

    f_Delay(10.0);

    //Release RRC security, SRBs and DRBs. Configure the SRBs and DRBs.

    f_NBIOT_SS_RRC_ReleaseSecurity(nbiot_Cell1);

    f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell1, USER_PLANE);

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1);    // @sic R5s180217 sic@

    //@siclog "Step 16" siclog@

    //The SS changes Ncell 1 levels according to row "T3" in Table 22.4.22.3.2-1

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 17" siclog@

    v_PhysCellId := f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell5); // @sic R5s170536 sic@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell5, nbiot_Cell1); // @sic R5s170536 sic@

    v_UE_Identity := cr_ReestabUE_Identity_NB(tsc_C_RNTI_Def, v_PhysCellId, v_ShortMAC_I);

    SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB(v_UE_Identity, otherFailure)));

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 17");

     //@siclog "Step 18-21 Void" siclog@

    //@siclog "Step 22" siclog@

    //The SS does not respond for 2.5 seconds (T301)

    t_Wait.start(f_NBIOT_SetTimerToleranceMin(rrcTimer, 2.5)); // @sic R5-176139 sic@

    alt { // @sic R5s180217 sic@

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB(v_UE_Identity, otherFailure)))

        {repeat;}

      [] t_Wait.timeout {}

    }

    f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell1, USER_PLANE);

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());    // @sic R5s180217 sic@

    //Activate RRC security

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(nbiot_Cell1, v_SecurityParams, rrcConnectionReestablishment);

    f_NBIOT_Security_Set(v_SecurityParams);

   //@siclog "Step 23-28" siclog@

    // Check: Does the UE perform TAU on Ncell1

    f_NBIOT_TrackingAreaUpdate(nbiot_Cell1, USER_PLANE);  // @sic R5-176139 sic@

    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, USER_PLANE, N1_IDLE); // @sic R5-176139 sic@

  }

}


After Change:

	function f_TC_22_4_22_NBIOT() runs on NBIOT_PTC

  {

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT1;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT2;

    var template (value) NB_CellPowerList_Type v_CellPowerList_AtT3;

    var SystemInformationBlockType2_NB_r13 v_SIB2;

    var EUTRA_ASN1_ShortMAC_I_Type v_ShortMAC_I;

    var EUTRA_ASN1_PhysCellId_Type v_PhysCellId;

    var template (present) EUTRA_ASN1_ReestabUE_Identity_Type v_UE_Identity;

    var NBIOT_SecurityParams_Type v_SecurityParams;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var boolean v_IPv4AllocationViaNasFlag := false;

    var template (omit) ExtdProtocolConfigOptions v_PcoToUE := omit;

    var template (omit) ExtdProtocolConfigOptions v_PcoFromUE := omit;

    var template (value) PDN_Address v_PDN_Address;

    var NBIOT_PDN_Type v_PDN;

    var GutiParameters_Type v_GutiParams;

    var template (value) MobileIdentity v_Guti;

    timer t_Wait;

    v_CellPowerList_AtT1 := {

      cs_NB_CellPower(nbiot_Cell1, tsc_NonSuitableOffNBIOTCellRS_EPRE),

      cs_NB_CellPower(nbiot_Cell5, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    v_CellPowerList_AtT2 := {

      cs_NB_CellPower(nbiot_Cell5, -85)

    };

    v_CellPowerList_AtT3 := {

      cs_NB_CellPower(nbiot_Cell1, -85),

      cs_NB_CellPower(nbiot_Cell5, tsc_NonSuitableOffNBIOTCellRS_EPRE)

    };

    f_NBIOT_Init(c2, USER_PLANE);  // @sic R5-176139 sic@

    //Set the value of T311 timer in SIB2 for Ncell 1, Ncell 5

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell1);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500; // @sic R5-176139 sic@

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms30000;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell1, v_SIB2);

    v_SIB2 := f_NBIOT_CellInfo_GetSIB2(nbiot_Cell5);

    v_SIB2.ue_TimersAndConstants_r13.t301_r13 := ms2500; // @sic R5-176139 sic@

    v_SIB2.ue_TimersAndConstants_r13.t311_r13 := ms30000;

    f_NBIOT_CellInfo_SetSIB2 (nbiot_Cell5, v_SIB2);

    //Create and configure all cells

    f_NBIOT_CellConfig_Def(nbiot_Cell1, USER_PLANE);

    f_NBIOT_CellConfig_Def(nbiot_Cell5, USER_PLANE); // @sic R5-176139 sic@

    //Bring UE to initial state @sic R5s170536 sic@

    f_NBIOT_SetCellPower(nbiot_Cell1, tsc_ServingCellRS_EPRE);

    // Switch on UE

    f_UT_SwitchOnUE(UT, false);

    v_GutiParams := f_NBIOT_CellInfo_GetGuti(nbiot_Cell1);

    v_Guti := f_GutiParameters2MobileIdentity(tsc_IEI_Guti, v_GutiParams);

    v_NAS_Ind := f_NBIOT_InitialRegistration_Step2_4 (nbiot_Cell1, USER_PLANE, PREAMBLE);

    f_NBIOT_InitialRegistration_Step5_12(nbiot_Cell1, USER_PLANE, STATE2_NB_CONNECTED_UPDATED, v_NAS_Ind);

    v_PDN := f_NBIOT_MobileInfo_GetAssignedPdn(0);  // This is registration, so this is the first PDN

    v_PcoFromUE := v_PDN.protocolConfigurationOptions;

    if (isvalue(v_PcoFromUE)){

      v_IPv4AllocationViaNasFlag := f_CheckExtdPCOforIPallocationViaNas(v_PcoFromUE);

    }

    v_PcoToUE := f_GetDefaultExtdProtocolConfigOptions(v_PcoFromUE);

    v_PDN_Address := f_CIOT_GetPDNAddress(v_PDN.assignedPdnType, v_IPv4AllocationViaNasFlag, -, pdnTypeAsSupportedByUE); // @sic R5s170597 sic@

    v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(tsc_SHT_IntegrityProtected_Ciphered,

                                                        cs_ATTACH_ACCEPT_Common(tsc_EAttachType_EpsOnly,

                                                                                cs_GprsTimer_v_deact,

                                                                                cs_TAIListNonConsecutive_lv(f_Asn2Nas_PlmnId (v_GutiParams.PLMN_Identity), {bit2oct(f_NBIOT_CellInfo_GetTAC(nbiot_Cell1))}),

                                                                                v_Guti,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                omit,

                                                                                cs_EPS_NwkFtSup_UP,

                                                                                f_GetNBIOTAdditionalUpdateResult(v_NAS_Ind.Pdu.Msg.aTTACH_REQUEST.addUpdateType, USER_PLANE),

                                                                                omit,

                                                                                omit),

                                                        cs_CIOT_ActivateDefEpsBearerContextRequest(tsc_EpsDefaultBearerId,

                                                                                                   v_PDN.pti,

                                                                                                   v_PDN.accessPointName,

                                                                                                   v_PDN_Address,

                                                                                                   v_PcoToUE,

                                                                                                   omit,

                                                                                                   omit));

    // Send specific message contents in RRC Connection Reconfiguration message

    SRB.send(cas_NB_SRB1_RrcNasPduList_REQ(nbiot_Cell1,

                                           cs_TimingInfo_Now,

                                           cs_RRCConnectionReconfiguration_NB(tsc_RRC_TI_Def, cs_RadioResourceConfigDedicated_NB_DrbConfig({ cs_508_NB_DRB_ToAddMod_DEFAULT(tsc_DRB1) },

                                                                                                                                           -,

                                                                                                                                           cs_RLF_TimersAndConstants_Setup_UP)),

                                           {v_NAS_MSG_Request}));

    SRB.receive(car_NB_SRB_RrcPdu_IND(nbiot_Cell1, tsc_SRB1, cr_RRCConnectionReconfigurationComplete_NB(tsc_RRC_TI_Def)));

    // Now finish preamble

    f_NBIOT_NAS_AttachComplete_UP(nbiot_Cell1, tsc_EpsDefaultBearerId);

    //WA#22_4_22

    f_NBIOT_SS_RRC_ReleaseSecurity(nbiot_Cell1);

    f_NBIOT_SS_SRBs_DRBs_Release(nbiot_Cell1, USER_PLANE, -, -, omit);
    f_NBIOT_TestBody_Set(true);

    //@siclog "Step 1" siclog@

    //The SS changes Ncell 1 levels according to row "T1" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 2" siclog@

    //Wait for 10 s.

    f_Delay(10.0);

    //@siclog "Step 3" siclog@

    //The SS changes Ncell 2 levels according to row "T2" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT2);

    //@siclog "Step 4-9" siclog@

    f_NBIOT_TrackingAreaUpdate(nbiot_Cell5, USER_PLANE,-,-,false); //WA#22_4_22
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 4");

    //@siclog "Step 10" siclog@

    //Wait for 5 s.

    f_Delay(5.0);

    //@siclog "Step 11" siclog@

    //The generic test procedure in TS 36.508 subclause 8.1.5A.2.3 is performed on Ncell 2

    f_NBIOT_GenericRbEst(nbiot_Cell5, mt_Access, -, -, cs_RLF_TimersAndConstants_Setup_UP); // @sic R5s170885 sic@

    //@siclog "Step 12" siclog@

    //The SS transmits an RRCConnectionReconfiguration-NB message on Ncell 2

    SRB.send(cas_NB_SRB_RrcPdu_REQ(nbiot_Cell5,

                                   tsc_SRB1,

                                   cs_TimingInfo_Now,

                                   cs_RRCConnectionReconfiguration_NB(tsc_RRC_TI_Def,

                                                                      cs_RadioResourceConfigDedicated_NB_DrbConfig(omit,

                                                                                                                   omit,

                                                                                                                   cs_RLF_TimersAndConstants_Release))));

    //@siclog "Step 13" siclog@

    SRB.receive(car_NB_SRB_RrcPdu_IND(nbiot_Cell5, tsc_SRB1, cr_RRCConnectionReconfigurationComplete_NB(tsc_RRC_TI_Def)));

    //@siclog "Step 14" siclog@

    //The SS changes Ncell 2 levels according to row "T1" in Table 22.4.22.3.2-1.

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT1);

    //@siclog "Step 15" siclog@

    //Wait for 10 s.

    f_Delay(10.0);

    //WA#22_4_22

    f_NBIOT_SS_SRBs_DRBs_Config(nbiot_Cell1, USER_PLANE, -, -, omit);
    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1);    // @sic R5s180217 sic@

    //@siclog "Step 16" siclog@

    //The SS changes Ncell 1 levels according to row "T3" in Table 22.4.22.3.2-1

    f_NBIOT_SetCellPowerList(v_CellPowerList_AtT3);

    //@siclog "Step 17" siclog@

    v_PhysCellId := f_NBIOT_CellInfo_GetPhyCellId(nbiot_Cell5); // @sic R5s170536 sic@

    v_ShortMAC_I := f_NBIOT_Calculate_ShortMAC(nbiot_Cell5, nbiot_Cell1); // @sic R5s170536 sic@

    v_UE_Identity := cr_ReestabUE_Identity_NB(tsc_C_RNTI_Def, v_PhysCellId, v_ShortMAC_I);

    SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB(v_UE_Identity, otherFailure)));

    f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell1, USER_PLANE); //WA#22_4_22
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 17");

     //@siclog "Step 18-21 Void" siclog@

    //@siclog "Step 22" siclog@

    //The SS does not respond for 2.5 seconds (T301)
    t_Wait.start(f_NBIOT_SetTimerToleranceMin(rrcTimer, 2.5)); // @sic R5-176139 sic@

    alt { // @sic R5s180217 sic@

      [] SRB.receive(car_NB_SRB0_RrcPdu_IND(nbiot_Cell1, cr_RRCConnectionReestablishmentRequest_NB(v_UE_Identity, otherFailure)))

        {

         f_NBIOT_SS_SRBs_DRBs_Reset(nbiot_Cell1, USER_PLANE); //WA#22_4_22
         repeat;

        }

      [] t_Wait.timeout {}

    }

    f_NBIOT_SS_ConfigRachProcedureMsg4(nbiot_Cell1, f_NBIOT_508_RRCConnectionSetup());    // @sic R5s180217 sic@

    //Activate RRC security

    v_SecurityParams := f_NBIOT_Security_Get();

    v_SecurityParams := f_NBIOT_SS_AS_ReestablishResumeSecurity(nbiot_Cell1, v_SecurityParams, rrcConnectionReestablishment);

    f_NBIOT_Security_Set(v_SecurityParams);

   //@siclog "Step 23-28" siclog@

    // Check: Does the UE perform TAU on Ncell1

    f_NBIOT_TrackingAreaUpdate(nbiot_Cell1, USER_PLANE,-,-,false);  // WA#22_4_22 @sic R5-176139 sic@
    f_NBIOT_PreliminaryPass(__FILE__, __LINE__, "Step 28");

    f_NBIOT_TestBody_Set(false);

    //Switch/power off UE

    f_NBIOT_Postamble(nbiot_Cell1, USER_PLANE, N1_IDLE); // @sic R5-176139 sic@
  }



1.2 f_NBIOT_SS_SRBs_DRBs_Release
	Function name
	f_NBIOT_SS_SRBs_DRBs_Release

	Reason for change
	DRB is not getting configured on Ncell 11 after the resumption of the RRC Connection. Hence, while releasing the RRC Connection and release of the SRBs and DRB locally, care should be taken to check if a DRB is configured or not

	Summary of change
	Introduced a flag to check for the presence of DRB before it is released. This change affects the following functions:

1. f_NBIOT_RRC_ConnectionRelease

2. f_NBIOT_RRC_ConnectionRelease_Common

3. f_NBIOT_RRC_ConnectionRelease_Local

4. f_NBIOT_TrackingAreaUpdate

Note : This change is covered by R5s180547 as well.

	TTCN module
	NBIOT_ConfigurationSteps.ttcn

	MCC160 Comment
	


Before Change:

	  function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,

                                        IOT_STATE_Type p_IOT_State,

                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,

                                        template (omit) boolean p_CnfFlag := omit) runs on NBIOT_PTC

  {

    var template (value) NB_RadioBearerList_Type v_RadioBearerList;

    var integer i;

    var integer k;

    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);

    if (p_IOT_State == USER_PLANE) {

      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);
      i := lengthof(v_RadioBearerList);

      for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

          v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);

        }

    }

    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);

  }



After Change:

	function f_NBIOT_SS_SRBs_DRBs_Release(NBIOT_CellId_Type p_CellId,

                                        IOT_STATE_Type p_IOT_State,

                                        template (value) TimingInfo_Type p_TimingInfo := cs_TimingInfo_Now,

                                        DRB_IdentityList_Type p_DrbList := tsc_DrbList_Def,

                                        template (omit) boolean p_CnfFlag := omit,

                                        boolean p_DrbConfigured := true /*WA#*/) runs on NBIOT_PTC

  {

    var template (value) NB_RadioBearerList_Type v_RadioBearerList;

    var integer i;

    var integer k;

    v_RadioBearerList[0] := cs_NB_SRB_Release(tsc_SRB1bis);

    if (p_IOT_State == USER_PLANE) {

      v_RadioBearerList[1] := cs_NB_SRB_Release(tsc_SRB1);

      if (p_DrbConfigured) {
          i := lengthof(v_RadioBearerList);

          for (k:=0; k < lengthof(p_DrbList); k:= k+1) {

            v_RadioBearerList[i+k] := cs_NB_DRB_Release(p_DrbList[k]);

          }

      }
    }

    f_NBIOT_SS_CommonRadioBearerConfig(p_CellId, v_RadioBearerList, p_TimingInfo, p_CnfFlag);

  }



